


8255 实验

一、实验内容


- 1、A 口工作在方式 0 输入方式，接 8 个开关，B 口工作在方式 0 输出方式，接 8 个 LED，**PC7 接单脉冲发生器，输入负脉冲时**，读一次 A 口，存储在 TABLE 数组单元中，同时在 B 口显示，读 10 次结束。
- 2、A 口工作在方式 0 输入方式，接 8 个开关，B 口工作在方式 0 输出方式，接 8 个 LED，**PC0 接产生周期 1s 脉冲的 8253，接收到 8253 输入每个负脉冲时**，读一次 A 口，存储在 TABLE 数组单元中，同时在 B 口显示，读 10 次结束。
- 3、A 口工作在方式 0，输入，接 8 个开关，B 口工作在方式 0，输出，接八段显示器;由 8253 产生 1S 的脉冲，PC7 接该脉冲，来个脉冲将读取 A 口，显示在屏幕上，同时;显示在八段显示器上，PC0, PC1 作为位码信号。
- 4、A 口工作在方式 0 输入方式，接 4 个开关，B 口工作在方式 0 输出方式，接 1 个七段显示器，**PC7 接单脉冲发生器，输入负脉冲时**，读一次 A 口，同时在 B 口显示，读 16 次结束。

二、芯片知识


1. 8255 方式控制字




8255 端口 C 按位置位/复位控制字




2. 8253 方式控制字




方式 0 波形



方式 2 波形




方式 3 波形



3. 实验台上的七段数码管为**共阴型**，**段码采用同相驱动**，输入端加**高电平**,**选中的数码管亮**，位码加反相驱动器，**位码输入端高电平选中**。

显示字形	g	e	f	d	c	b	a	段码
0	0	1	1	1	1	1	1	3fh
1	0	0	0	0	1	1	0	06h
2	1	0	1	1	0	1	1	5bh
3	1	0	0	1	1	1	1	4fh
4	1	1	0	0	1	1	0	66h
5	1	1	0	1	1	0	1	6dh
6	1	1	1	1	1	0	1	7dh
7	0	0	0	0	1	1	1	07h
8	1	1	1	1	1	1	1	7fh
9	1	1	0	1	1	1	1	6fh



三、参考程序

1. 8255 A 口, 工作在方式 0, 输入, 接 8 个开关, B 口方式 0, 输出, 接 8 个 LED, PC7 接单脉冲, 输入负脉冲时, 读一次 A 口, 存入 TABLE 内存单元中, 并在 B 口显示。

```
DATA    SEGMENT
    PA    EQU    288H
    PB    EQU    289H
    PC    EQU    28AH
    CTL55 EQU    28BH
    TABLE DB    10 DUP(?)
DATA    ENDS

STACK1  SEGMENT PARA STACK
        DW 20H DUP(0)
STACK1  ENDS

CODE    SEGMENT
        ASSUME CS: CODE, DS: DATA
START:
        MOV AX, DATA
        MOV DS, AX
        MOV DX, CTL55
        MOV AL, 10011000B
        OUT DX, AL
        MOV CX, 10
        MOV SI, 0

WAT:
        MOV DX, PC ;读 C 口
        IN  AL, DX
        AND AL, 80H ;80H=1000 0000B 检测 PC7
        JNZ WAT ;

        MOV DX, PA ;PC7=0, 读 A 口
        IN  AL, DX
        MOV DX, PB ;送 B 口
        OUT DX, AL
        MOV TABLE[SI], AL ;将读取的开关值存放到内存单元 TABLE 中
        INC SI

WAT1:
        MOV DX, PC ;读 C 口
        IN  AL, DX
        AND AL, 80H ;检测 PC7
        JZ  WAT1 ;
        LOOP WAT
```

```

        MOV AH,4CH
        INT 21H
CODE    ENDS
        END START

```

3. ;A 口工作在方式 0，输入，接 8 个开关，B 口工作在方式 0，输出，接八段显示器；
 ;由 8253 产生 1S 的脉冲，PC7 接该脉冲，来个脉冲将读取 A 口，显示在屏幕上，同时
 ;显示在八段显示器上，PC0, PC1 作为位码信号。

```

DATA SEGMENT
    T0 EQU 280H    ;T0---CTL 为计数器 0---控制寄存器的地址
    T1 EQU 281H
    T2 EQU 282H
    CTL1 EQU 283H
    PA EQU 288H
    PB EQU 289H
    PC EQU 28AH
    CTL EQU 28BH
    LTABLE DB 3FH, 06H, 5BH, 4FH, 66H, 6DH, 7DH, 07H
           DB 7FH, 6FH, 77H, 7CH, 39H, 5EH, 79H, 71H
    DAT DB 21H
DATA ENDS

```

```

CODE SEGMENT
    ASSUME CS:CODE,DS:DATA
;*****主程序*****
START:
    MOV AX, DATA
    MOV DS, AX

    CALL Init_8253;
    CALL Init_8255;
    AGAIN:
        CALL READ_PA
        CALL DISP_DAT
        JMP AGAIN

    MOV AH,4CH
    INT 21H
;*****主程序结束*****
;*****数码管显示字符程序*****
LED_1 PROC    ;显示字符程序
    LEA SI, LTABLE

```

```

        ADD     SI, BX
        MOV     AL,[SI]
        AND     AL,7FH      ;最高位设为 0，以不显示小数点
        MOV     DX,PB
        OUT     DX,AL
        RET
LED_1   ENDP

LED_S   PROC
        PUSH    CX
        PUSH    BX

        MOV     DX, PC      ;悬位码
        MOV     AL, 0H
        OUT     DX, AL

        MOV     BH, 0
        AND     BL, 0FH     ;显示低四位
        CALL    LED_1
        MOV     DX, PC      ;选择低四位,即 PC0 置为 1
        MOV     AL, 1H
        OUT     DX, AL

        MOV     DX, PC      ; 熄位码
        MOV     AL, 0H
        OUT     DX, AL

        POP     BX
        PUSH    BX

        MOV     BH, 0
        MOV     CL, 4
        SHR     BL, CL
        AND     BL, 0FH     ;显示高四位
        CALL    LED_1
        MOV     DX, PC      ;选择高四位，即 PC1 置为 1
        MOV     AL, 02H
        OUT     DX, AL

        MOV     DX,PC      ;熄位码
        MOV     AL,0H
        OUT     DX,AL

        POP     BX

```

```

        POP        CX
        RET
LED_S   ENDP
;*****读取 A 口数据*****
READ_PA PROC
    PUSH DX
    PUSH AX

LOOP_CHECK:
    MOV  BH, 0
    MOV  BL, DAT
    CALL LED_S
    MOV  DX, PC    ;8255 C 口 读入
    IN   AL, DX
    TEST AL, 80H   ; 80H=1000 0000B 判断 1S 脉冲
    JNZ  LOOP_CHECK
    MOV  DX, PA
    IN   AL, DX
    MOV  DAT, AL

    POP  AX
    POP  DX
    ret
READ_PA ENDP

;*****电脑终端显示字符程序*****
DISP_DAT PROC
    PUSH DX
    PUSH AX
    PUSH CX

    MOV  AL, DAT
    MOV  CL, 4
    SHR  AL, CL
    AND  AL, 0FH

    CMP  AL, 0AH
    JC   S1
    ADD  AL, 7H
S1:
    ADD  AL, 30H
    MOV  DL, AL    ;高四位显示在屏幕上
    MOV  AH, 02H
    INT  21H

```

```

MOV AL, DAT
AND AL, 0FH
CMP AL, 0AH
JC S2
ADD AL, 7H
S2:
ADD AL, 30H
MOV DL, AL ;低四位显示在屏幕上
MOV AH, 02H
INT 21H

MOV DL, '' ;显示空格
MOV AH, 02H
INT 21H

POP CX
POP AX
POP DX
ret
DISP_DAT ENDP

;*****初始化 8253*****
Init_8253 PROC;
PUSH DX
PUSH AX
; /*8253 产生 1S 的连续方波
MOV DX, CTL1
MOV AL, 00100101B ;选择通道 0, 写入高字节, 方式 2, 用 BCD 码
OUT DX, AL
MOV DX, T0
MOV AL, 20H
OUT DX, AL
;设置第二级
MOV DX, CTL1
MOV AL, 01100111B ;选择通道 1, 写入高字节, 方式 3, 用 BCD 码
OUT DX, AL
MOV DX, T1
MOV AL, 10H
OUT DX, AL
;8253 产生 1S 的连续方波*/

POP AX
POP DX

```



```

        RET
Init_8253 ENDP
;*****初始化 8255*****
Init_8255 PROC;
    PUSH DX
    PUSH AX

    MOV DX,CTL ;8255 方式字控制 ;10011000
    MOV AL, 98H
    OUT DX, AL

    POP AX
    POP DX
    RET
Init_8255 ENDP

CODE ENDS
    END START

```

四、参照三完成题目 2， 4